Example 9a: Multivariate Piecewise Slope Models
(complete data, syntax, and output available for SAS, and Mplus MLM/SEM electronically)

This example uses the same data as in Hoffman (2015) chapter 6. These data are from a short-term longitudinal study of six occasions over 2 weeks for 101 adults age 65–80 years. We will see how performance on two versions of a processing speed task (called “number match 3” and “number match 5”), as measured by response time in milliseconds / 10, *differentially* declines (improves) over the 6 practice sessions, as well as to what extent baseline age *differentially* predicts these differences. In this example we will use piecewise models of change, in which one slope captures change from sessions 1–2, and another captures change from sessions 2–6 (as were previously examined for nm3rt in CLDP 944 Example 6). We are using ML so that the results from SAS are as close as possible to those of Mplus, other than denominator degrees of freedom (which are not used in Mplus, but for which we selected Satterthwaite as usual in SAS). Rather than using the DV3 and DV5 dummy codes, we are letting the CLASS statement make them for us by using DV as a categorical predictor, which results in less code (but otherwise equivalent models). <u>The key to this direct, DV-specific interpretation is to omit the general fixed intercept and any predictor main effects.</u>

SAS Code for Data Manipulation and Creating Mplus Stacked Data File for Multivariate MLM:

```
* Location of original data files - CHANGE THIS;
%LET filesave= C:\Dropbox\_Archive\Example Data\945 Multivariate MLM\Practice Example;
LIBNAME filesave "&filesave.";

* Example data: DVs = nm3rt, nm5rt, IVs = session, age, educyrs;
* Bring data file into work library;
DATA work.Practice; SET filesave.PracticeEffects;
* Scaling outcomes in deci-seconds -- Mplus threw up otherwise;
  nm3rt = nm3rt / 100;     LABEL nm3rt= "Number Match 3 RT per 100 ms";
  nm5rt = nm5rt / 100;     LABEL nm5rt= "Number Match 5 RT per 100 ms";
* Creating two slopes for piecewise models;
     IF session = 1 THEN DO; slope12 = 0; slope26 = 0; END;
ELSE IF session = 2 THEN DO; slope12 = 1; slope26 = 0; END;
ELSE IF session > 2 THEN DO; slope12 = 1; slope26 = session-2; END;
LABEL slope12 = "1-2 Early Practice Slope"
      slope26 = "2-6 Later Practice Slope";
* Center time-invariant age;
  age80 = age - 80; LABEL age80 = "Age (0=80)";
* Rename ID for clarity;
  RENAME ID=PersonID; RUN;
PROC SORT DATA=work.Practice; BY PersonID; RUN;

* Trimming data to send just needed variables to Mplus;
DATA work.PracticeMplus; * RETAIN re-orders variables as listed;
     RETAIN PersonID nm3rt nm5rt session slope12 slope26 age80;
     SET work.Practice;
     * Telling it which variables to keep -- handy to use this in Mplus;
     KEEP PersonID nm3rt nm5rt session slope12 slope26 age80;
     * Replace any missing values with -999;
     ARRAY avars(7) PersonID nm3rt nm5rt session slope12 slope26 age80;;
     DO i=1 TO 7; IF avars(i)=. THEN avars(i)=-999; END; DROP i;
RUN;

* Export to .csv for use in Mplus MLM syntax;
PROC EXPORT DATA=work.PracticeMplus OUTFILE= "&filesave.\practice.csv"
      DBMS=CSV REPLACE; PUTNAMES=NO; RUN;

* Stack data for multivariate models in SAS;
DATA work.PracticeMultiv; SET work.Practice;
* DV will be used ON CLASS, y is outocme, dv3 and dv5 are "switches";
  DV="nm3rt"; y=nm3rt; dv3=1; dv5=0; OUTPUT;
  DV="nm5rt"; y=nm5rt; dv3=0; dv5=1; OUTPUT;
RUN;
```

Model 1. Multivariate Empty Means, Random Intercepts (for *t* = time, *i* = individual, *d* = DV)

Level 1: $y_{tid} = DV3[\beta_{0i3} + e_{ti3}] + DV5[\beta_{0i5} + e_{ti5}]$

Level 2: Intercepts: $\beta_{0i3} = \gamma_{003} + U_{0i3}$

$\beta_{0i5} = \gamma_{005} + U_{0i5}$

These equations will use the DV3 and DV5 dummy codes (as created during the previous multivariate stacking DATA step) to act as "switches" such that they control which model parameters are used to predict each row of data. The model then predicts each DV separately but simultaneously, which allows us to estimate covariances among the residuals and random effects, as well as test differences in fixed effects across DVs..

```
TITLE "Model 1: Empty Means, Multivariate Random Intercept Model";
PROC MIXED DATA=work.PracticeMultiv COVTEST NOCLPRINT NAMELEN=100 IC METHOD=ML;
     CLASS PersonID session DV; * NOINT removes general intercept;
     MODEL y = DV / NOINT SOLUTION DDFM=Satterthwaite;
     RANDOM    DV / G GCORR V VCORR TYPE=UN SUBJECT=PersonID; * Level 2;
     REPEATED  DV / R RCORR TYPE=UN SUBJECT=PersonID*session; * Level 1;
     ODS OUTPUT CovParms=CovEmpty; * Save for pseudo-R2; RUN;
```

```
             Dimensions
Covariance Parameters           24
Columns in X                     6
Columns in Z per Subject         6
Subjects                       101
Max Obs per Subject             12

   Estimated R Matrix for              Estimated R Correlation
   PersonID*session 101 1           Matrix for PersonID*session 101 1
 Row        Col1        Col2         Row        Col1        Col2
   1      4.4900      3.3189           1      1.0000      0.4073
   2      3.3189     14.7854           2      0.4073      1.0000

                         Estimated G Matrix
                              Participant
 Row    Effect    DV        ID                  Col1        Col2
   1    DV        nm3rt            101       19.8820     39.2314
   2    DV        nm5rt            101       39.2314     96.0023

                   Estimated G Correlation Matrix
                              Participant
 Row    Effect    DV        ID                  Col1        Col2
   1    DV        nm3rt            101        1.0000      0.8980
   2    DV        nm5rt            101        0.8980      1.0000
```

The **R** matrix gives the level-1 residual variance for each DV (in order alphabetically or numerically), as well as their covariance. The **RCORR** matrix gives the level-1 correlation among the residuals across DVs: after controlling for individual mean differences (via the level-2 random intercepts), if you are higher than usual on DV3 at a given occasion, are you also higher than usual on DV5 at that same occasion?

The **G** matrix gives the level-2 random intercept variances for each DV, as well as their covariance. The **GCORR** matrix gives the level-2 correlation among the random intercepts across DVs: if you are higher than others on average on DV3, are you also higher than others on average on DV5?

Because all variance model parameters are estimated separately per DV, each gets its own ICC as well:
DV3 ICC = 19.88 / (19.88+ 4.49) = .816
DV5 ICC = 96.00 / (96.00+ 14.79) = .755

```
                     Covariance Parameter Estimates
                                              Standard         Z
Cov Parm     Subject              Estimate       Error     Value        Pr Z
UN(1,1)      PersonID              19.8820      2.9035      6.85      <.0001  L2 DV3 intercept variance
UN(2,1)      PersonID              39.2314      5.9823      6.56      <.0001  L2 intercept covariance
UN(2,2)      PersonID              96.0023     13.8570      6.93      <.0001  L2 DV5 intercept variance
UN(1,1)      PersonID*session       4.4900      0.2826     15.89      <.0001  L1 DV3 residual variance
UN(2,1)      PersonID*session       3.3189      0.3915      8.48      <.0001  L1 residual covariance
UN(2,2)      PersonID*session      14.7854      0.9305     15.89      <.0001  L1 DV5 residual variance

                            Information Criteria
Neg2LogLike     Parms        AIC        AICC        HQIC         BIC        CAIC
     6445.3         8     6461.3      6461.4      6469.7      6482.2      6490.2

                      Solution for Fixed Effects
                                   Standard
Effect     DV        Estimate        Error       DF     t Value     Pr > |t|
DV         nm3rt      17.7070       0.4520      101       39.18       <.0001  DV3 fixed intercept
DV         nm5rt      34.5174       0.9874      101       34.96       <.0001  DV5 fixed intercept
```

Model 2. Multivariate Fixed Piecewise Slopes, Random Intercepts

$$\text{Level 1: } y_{tid} = DV3\left[\beta_{0i3} + \beta_{1i3}(Slope12_{ti3}) + \beta_{2i3}(Slope26_{ti3}) + e_{ti3}\right] + DV5\left[\beta_{0i5} + \beta_{1i5}(Slope12_{ti5}) + \beta_{2i5}(Slope26_{ti5}) + e_{ti5}\right]$$

$$\text{Level 2: Intercepts: } \beta_{0i3} = \gamma_{003} + U_{0i3} \qquad \beta_{0i5} = \gamma_{005} + U_{0i5}$$

$$\text{Slope12: } \beta_{1i3} = \gamma_{103} \qquad \beta_{1i5} = \gamma_{105}$$

$$\text{Slope26: } \beta_{2i3} = \gamma_{203} \qquad \beta_{2i5} = \gamma_{205}$$

Here we add separate beta placeholders at level 1 for each piecewise slope for each DV, each of which are then defined at level 2 with fixed effects only (for now).

```
TITLE1 "Model 2: Fixed Piecewise Slopes, Random Intercept Multivariate Model";
PROC MIXED DATA=work.PracticeMultiv COVTEST NOCLPRINT NAMELEN=100 IC METHOD=ML;
     CLASS PersonID session DV; * Note: no general fixed slope main effects;
     MODEL y = DV DV*slope12 DV*slope26 / NOINT SOLUTION DDFM=Satterthwaite OUTPM=work.PredTime;
     RANDOM    DV / G GCORR TYPE=UN SUBJECT=PersonID;           * Level 2;
     REPEATED  DV / R RCORR TYPE=UN SUBJECT=PersonID*session;  * Level 1;
     ODS OUTPUT InfoCrit=FitFixed CovParms=CovFixed; * Save for LRT, pseudo-R2; RUN;
```

Estimated R Matrix for PersonID*session 101 1

Row	Col1	Col2
1	3.3963	2.3138
2	2.3138	13.7255

Estimated R Correlation Matrix for PersonID*session 101 1

Row	Col1	Col2
1	1.0000	0.3389
2	**0.3389**	1.0000

This residual **R** matrix covariance is now controlling for the fixed slopes.

Estimated G Matrix

Row	Effect	DV	Participant ID	Col1	Col2
1	DV	nm3rt	101	20.0643	39.3989
2	DV	nm5rt	101	39.3989	96.1789

Estimated G Correlation Matrix

Row	Col1	Col2
1	1.0000	0.8969
2	**0.8969**	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
UN(1,1)	PersonID	20.0643	2.9033	6.91	<.0001	L2 DV3 intercept variance
UN(2,1)	PersonID	39.3989	5.9822	6.59	**<.0001**	**L2 intercept covariance**
UN(2,2)	PersonID	96.1789	13.8569	6.94	<.0001	L2 DV5 intercept variance
UN(1,1)	PersonID*session	3.3963	0.2137	15.89	<.0001	L1 DV3 residual variance
UN(2,1)	PersonID*session	2.3138	0.3208	7.21	**<.0001**	**L1 residual covariance**
UN(2,2)	PersonID*session	13.7255	0.8638	15.89	<.0001	L1 DV5 residual variance

Solution for Fixed Effects

Effect	DV	Estimate	Standard Error	DF	t Value	Pr > |t|	
DV	nm3rt	19.6189	0.4820	130	40.71	<.0001	DV3 fixed intercept
DV	nm5rt	36.6095	1.0432	125	35.10	<.0001	DV5 fixed intercept
slope12*DV	nm3rt	-1.6364	0.2320	505	-7.06	<.0001	DV3 fixed slope12
slope12*DV	nm5rt	-2.3734	0.4663	505	-5.09	<.0001	DV5 fixed slope12
slope26*DV	nm3rt	-0.3289	0.05799	505	-5.67	<.0001	DV3 fixed slope26
slope26*DV	nm5rt	-0.06859	0.1166	505	-0.59	0.5565	DV5 fixed slope26

```
* Pseudo-R2 for fixed slopes relative to empty model;
%PseudoR2(Ncov=6, CovFewer=CovEmpty, CovMore=CovFixed);
```

PsuedoR2 (% Reduction) for CovEmpty vs. CovFixed

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2	
CovEmpty	UN(1,1)	PersonID	19.8820	2.9035	6.85	<.0001	.	
CovEmpty	UN(2,2)	PersonID	96.0023	13.8570	6.93	<.0001	.	
CovEmpty	UN(1,1)	PersonID*session	4.4900	0.2826	15.89	<.0001	.	
CovEmpty	UN(2,2)	PersonID*session	14.7854	0.9305	15.89	<.0001	.	
CovFixed	UN(1,1)	PersonID	20.0643	2.9033	6.91	<.0001	-0.00917	for DV3 int
CovFixed	UN(2,2)	PersonID	96.1789	13.8569	6.94	<.0001	-0.00184	for DV5 int
CovFixed	UN(1,1)	PersonID*session	3.3963	0.2137	15.89	<.0001	**0.24359**	for DV3 res
CovFixed	UN(2,2)	PersonID*session	13.7255	0.8638	15.89	<.0001	**0.07169**	for DV3 res

Model 3. Multivariate Fixed Piecewise Slopes, Random Intercepts

$$\text{Level 1: } y_{tid} = DV3\left[\beta_{0i3} + \beta_{1i3}(Slope12_{ti3}) + \beta_{2i3}(Slope26_{ti3}) + e_{ti3}\right] + DV5\left[\beta_{0i5} + \beta_{1i5}(Slope12_{ti5}) + \beta_{2i5}(Slope26_{ti5}) + e_{ti5}\right]$$

$$\text{Level 2: Intercepts: } \beta_{0i3} = \gamma_{003} + U_{0i3} \qquad \beta_{0i5} = \gamma_{005} + U_{0i5}$$

$$\text{Slope12: } \beta_{1i3} = \gamma_{103} + U_{1i3} \qquad \beta_{1i5} = \gamma_{105} + U_{1i5}$$

$$\text{Slope26: } \beta_{2i3} = \gamma_{203} + U_{2i3} \qquad \beta_{2i5} = \gamma_{205} + U_{2i5}$$

Here we add four random slopes, one for each slope and DV. We are doing this for both DVs at once here for expediency, but in practice you could (and pry should) test each random slope separately as usual.

```
TITLE1 "Model 3: Multivariate Random Piecewise Slopes Model";
PROC MIXED DATA=work.PracticeMultiv COVTEST NOCLPRINT NAMELEN=100 IC METHOD=ML;
     CLASS PersonID session DV; * Note: also no general random slope main effects;
     MODEL y = DV DV*slope12 DV*slope26 / NOINT SOLUTION DDFM=Satterthwaite;
     RANDOM    DV DV*slope12 DV*slope26 / G GCORR TYPE=UN SUBJECT=PersonID; * Level 2;
     REPEATED  DV / R RCORR TYPE=UN SUBJECT=PersonID*session; * Level 1;
     ODS OUTPUT InfoCrit=FitRand CovParms=CovRand; * Save for LRT, pseudo-R2;
     ESTIMATE "DV Diff in Intercept" DV -1 1;
     ESTIMATE "DV Diff in Slope12"   DV*slope12 -1 1;
     ESTIMATE "DV Diff in Slope26"   DV*slope26 -1 1;
     CONTRAST "DV Diff in Both Slopes"  DV*slope12 -1 1, DV*slope26 -1 1 / CHISQ; RUN;
```

Estimated R Matrix for PersonID*session 101 1

Row	Col1	Col2
1	1.7673	1.2799
2	1.2799	10.8625

Estimated R Correlation Matrix for PersonID*session 101 1

Row	Col1	Col2
1	1.0000	0.2921
2	**0.2921**	1.0000

This residual **R** matrix covariance is now controlling for the fixed and random slopes.

Estimated G Matrix

Row	Effect	DV	Participant ID	Col1	Col2	Col3	Col4	Col5	Col6
1	DV	nm3rt	101	28.1322	48.8623	-5.3558	-6.9889	-1.0538	-0.9163
2	DV	nm5rt	101	48.8623	130.24	-2.0792	-21.6130	-1.9294	-2.2901
3	slope12*DV	nm3rt	101	-5.3558	-2.0792	6.3041	1.6244	-0.1621	0.3584
4	slope12*DV	nm5rt	101	-6.9889	-21.6130	1.6244	7.8667	0.6837	1.4381
5	slope26*DV	nm3rt	101	-1.0538	-1.9294	-0.1621	0.6837	0.2573	0.1559
6	slope26*DV	nm5rt	101	-0.9163	-2.2901	0.3584	1.4381	0.1559	0.2224

The **GCORR** matrix provides all possible covariances among the random intercepts and slopes (between DVs are in bold).

Estimated G Correlation Matrix

Row	Effect	DV	Participant ID	Col1	Col2	Col3	Col4	Col5	Col6
1	DV	nm3rt	101	1.0000	0.8072	-0.4022	-0.4698	-0.3917	-0.3663
2	DV	nm5rt	101	**0.8072**	1.0000	-0.07256	-0.6752	-0.3333	-0.4255
3	slope12*DV	nm3rt	101	-0.4022	-0.07256	1.0000	0.2307	-0.1273	0.3027
4	slope12*DV	nm5rt	101	-0.4698	-0.6752	**0.2307**	1.0000	0.4805	1.0000
5	slope26*DV	nm3rt	101	-0.3917	-0.3333	-0.1273	0.4805	1.0000	0.6517
6	slope26*DV	nm5rt	101	-0.3663	-0.4255	0.3027	**1.0000**	**0.6517**	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
UN(1,1)	PersonID	28.1322	4.2099	6.68	<.0001	L2 DV3 intercept variance
UN(2,1)	PersonID	48.8623	8.1690	5.98	**<.0001**	**L2 intercept covariance**
UN(2,2)	PersonID	130.24	19.8749	6.55	<.0001	L2 DV5 intercept variance
UN(3,1)	PersonID	-5.3558	1.7962	-2.98	0.0029	
UN(3,2)	PersonID	-2.0792	3.5969	-0.58	0.5632	
UN(3,3)	PersonID	6.3041	1.3054	4.83	<.0001	L2 DV3 slope12 variance
UN(4,1)	PersonID	-6.9889	2.8670	-2.44	0.0148	
UN(4,2)	PersonID	-21.6130	6.8184	-3.17	0.0015	
UN(4,3)	PersonID	1.6244	1.6100	1.01	**0.3130**	**L2 slope12 covariance**
UN(4,4)	PersonID	7.8667	3.8230	2.06	0.0198	L2 DV5 slope12 variance

UN(5,1)	PersonID	-1.0538	0.3734	-2.82	0.0048	
UN(5,2)	PersonID	-1.9294	0.8019	-2.41	0.0161	
UN(5,3)	PersonID	-0.1621	0.2066	-0.78	0.4328	
UN(5,4)	PersonID	0.6837	0.3362	2.03	0.0420	
UN(5,5)	PersonID	0.2573	0.06273	4.10	<.0001	L2 DV3 slope26 variance
UN(6,1)	PersonID	-0.9163	0.6291	-1.46	0.1452	
UN(6,2)	PersonID	-2.2901	1.3712	-1.67	0.0949	
UN(6,3)	PersonID	0.3584	0.3481	1.03	0.3032	
UN(6,4)	PersonID	1.4381	0.6030	2.38	0.0171	
UN(6,5)	PersonID	0.1559	0.08431	1.85	**0.0644**	**L2 slope26 covariance**
UN(6,6)	PersonID	0.2224	0.2042	1.09	0.1380	L2 DV5 slope26 variance
UN(1,1)	PersonID*session	1.7673	0.1436	12.31	<.0001	L1 DV3 residual variance
UN(2,1)	PersonID*session	1.2799	0.2622	4.88	**<.0001**	**L1 residual covariance**
UN(2,2)	PersonID*session	10.8625	0.8825	12.31	<.0001	L1 DV5 residual variance

Solution for Fixed Effects

Effect	DV	Estimate	Standard Error	DF	t Value	Pr > |t|	
DV	nm3rt	19.6189	0.5441	101	36.06	<.0001	DV3 fixed intercept
DV	nm5rt	36.6095	1.1820	101	30.97	<.0001	DV5 fixed intercept
slope12*DV	nm3rt	-1.6364	0.3007	101	-5.44	<.0001	DV3 fixed slope12
slope12*DV	nm5rt	-2.3734	0.5000	101	-4.75	<.0001	DV5 fixed slope12
slope26*DV	nm3rt	-0.3289	0.06555	101	-5.02	<.0001	DV3 fixed slope26
slope26*DV	nm5rt	-0.06859	0.1138	101	-0.60	0.5481	DV5 fixed slope26

Type 3 Tests of Fixed Effects

In using DV on the CLASS statement, SAS automatically creates these multivariate Wald tests for the combined effect of each predictor across DVs.

Effect	Num DF	Den DF	F Value	Pr > F	
DV	2	101	662.28	<.0001	Are both intercepts NE 0?
slope12*DV	2	101	21.06	<.0001	Are both fixed slope12 NE 0?
slope26*DV	2	101	13.55	<.0001	Are both fixed slope26 NE 0?

Estimates

As we requested, these ESTIMATE statements test the difference across DVs in each fixed effect. Now we have all possible relevant pieces of information about the fixed effects from the same model!

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
DV Diff in Intercept	16.9905	0.8367	101	20.31	<.0001
DV Diff in Slope12	-0.7369	0.5174	101	-1.42	0.1574
DV Diff in Slope26	0.2603	0.1079	101	2.41	0.0176

Contrasts

As we requested, this CONTRAST statement tests multiple DV differences at once (here, in the two slopes simultaneously).

Label	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
DV Diff in Both Slopes	2	101	6.31	3.16	0.0426	0.0468

```
* LRT against fixed slopes model;
%FitTest(FitFewer=FitFixed, FitMore=FitRand);
```

Do the four new random slopes (and their 14 new covariances) significantly improve model fit?
Yes, $-2\Delta LL(\sim 18) = 149.60$, $p < .001$

Likelihood Ratio Test for FitFixed vs. FitRand

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitFixed	6296.7	12	6320.7	6352.1	.	.	.
FitRand	6147.1	30	6207.1	6285.6	149.600	18	0

Model 4. Add Age Predicting Multivariate Random Piecewise Slopes

Level 1: $y_{tid} = DV3[\beta_{0i3} + \beta_{1i3}(Slope12_{ti3}) + \beta_{2i3}(Slope26_{ti3}) + e_{ti3}] +$

$DV5[\beta_{0i5} + \beta_{1i5}(Slope12_{ti5}) + \beta_{2i5}(Slope26_{ti5}) + e_{ti5}]$

Level 2: Intercepts: $\beta_{0i3} = \gamma_{003} + \gamma_{013}(Age_i - 80) + U_{0i3}$ $\beta_{0i5} = \gamma_{005} + \gamma_{015}(Age_i - 80) + U_{0i5}$

Slope12: $\beta_{1i3} = \gamma_{103} + \gamma_{113}(Age_i - 80) + U_{1i3}$ $\beta_{1i5} = \gamma_{105} + \gamma_{115}(Age_i - 80) + U_{1i5}$

Slope26: $\beta_{2i3} = \gamma_{203} + \gamma_{213}(Age_i - 80) + U_{2i3}$ $\beta_{2i5} = \gamma_{205} + \gamma_{215}(Age_i - 80) + U_{2i5}$

Here we add six fixed effects of age: predicting the random intercept and each random piecewise slopes for each DV. In order to test if age predicts each DV differently, they must be part of the same model—multivariate is the only way!

```
TITLE1 "Model 4: Add Age Predicting Multivariate Random Piecewise Slopes";
PROC MIXED DATA=work.PracticeMultiv COVTEST NOCLPRINT NAMELEN=100 IC METHOD=ML;
     CLASS PersonID session DV;
     MODEL y = DV DV*slope12 DV*slope26 DV*age80 DV*slope12*age80 DV*slope26*age80
               / NOINT SOLUTION DDFM=Satterthwaite OUTPM=PredAge;
     RANDOM    DV DV*slope12 DV*slope26 / G GCORR TYPE=UN SUBJECT=PersonID; * Level 2;
     REPEATED  DV / R RCORR TYPE=UN SUBJECT=PersonID*session; * Level 1;
     ODS OUTPUT InfoCrit=FitAge CovParms=CovAge; * Save for LRT, pseudo-R2;
     ESTIMATE "DV Diff in Age Main Effect" DV*age80 -1 1;
     ESTIMATE "DV Diff in Age*Slope12"     DV*slope12*age80 -1 1;
     ESTIMATE "DV Diff in Age*Slope26"     DV*slope26*age80 -1 1;
CONTRAST "DV3 DF=3 age effect"       DV*age80  1 0, DV*slope12*age80  1 0, DV*slope26*age80  1 0 / CHISQ;
CONTRAST "DV5 DF=3 age effect"       DV*age80  0 1, DV*slope12*age80  0 1, DV*slope26*age80  0 1 / CHISQ;
CONTRAST "DV Diff in 3 Age Effects"  DV*age80 -1 1, DV*slope12*age80 -1 1, DV*slope26*age80 -1 1 / CHISQ;
RUN;
```

Estimated R Matrix for PersonID*session 101 1

Row	Col1	Col2
1	1.7673	1.2799
2	1.2799	10.8625

Estimated R Correlation Matrix for PersonID*session 101 1

Row	Col1	Col2
1	1.0000	0.2921
2	**0.2921**	1.0000

Estimated G Matrix

Row	Effect	DV	Participant ID	Col1	Col2	Col3	Col4	Col5	Col6
1	DV	nm3rt	101	24.8900	44.1108	-4.5304	-7.1420	-0.9069	-0.7587
2	DV	nm5rt	101	44.1108	123.27	-0.8696	-21.8373	-1.7140	-2.0591
3	slope12*DV	nm3rt	101	-4.5304	-0.8696	6.0940	1.6634	-0.1995	0.3183
4	slope12*DV	nm5rt	101	-7.1420	-21.8373	1.6634	7.8595	0.6906	1.4456
5	slope26*DV	nm3rt	101	-0.9069	-1.7140	-0.1995	0.6906	0.2507	0.1488
6	slope26*DV	nm5rt	101	-0.7587	-2.0591	0.3183	1.4456	0.1488	0.2147

Estimated G Correlation Matrix

Row	Effect	DV	Participant ID	Col1	Col2	Col3	Col4	Col5	Col6
1	DV	nm3rt	101	1.0000	0.7963	-0.3679	-0.5106	-0.3630	-0.3282
2	DV	nm5rt	101	**0.7963**	1.0000	-0.03173	-0.7016	-0.3083	-0.4002
3	slope12*DV	nm3rt	101	-0.3679	-0.03173	1.0000	0.2404	-0.1614	0.2782
4	slope12*DV	nm5rt	101	-0.5106	-0.7016	**0.2404**	1.0000	0.4920	1.0000
5	slope26*DV	nm3rt	101	-0.3630	-0.3083	-0.1614	0.4920	1.0000	0.6411
6	slope26*DV	nm5rt	101	-0.3282	-0.4002	0.2782	**1.0000**	**0.6411**	1.0000

These **GCORR** L2 correlations now control for age.

Solution for Fixed Effects

Effect	DV	Estimate	Standard Error	DF	t Value	Pr > |t|	
DV	nm3rt	19.6686	0.5139	101	38.27	<.0001	DV3 fixed intercept
DV	nm5rt	36.6822	1.1529	101	31.82	<.0001	DV5 fixed intercept
slope12*DV	nm3rt	-1.6491	0.2973	101	-5.55	<.0001	DV3 fixed slope12
slope12*DV	nm5rt	-2.3710	0.5001	101	-4.74	<.0001	DV5 fixed slope12
slope26*DV	nm3rt	-0.3312	0.06508	101	-5.09	<.0001	DV3 fixed slope26
slope26*DV	nm5rt	-0.07100	0.1135	101	-0.63	0.5331	DV5 fixed slope26

```
age80*DV           nm3rt      0.2978     0.08497     101        3.50       0.0007  DV3 age on intercept
age80*DV           nm5rt      0.4364      0.1906     101        2.29       0.0241  DV5 age on intercept
slope12*age80*DV   nm3rt    -0.07581     0.04916     101       -1.54       0.1261  DV3 age on slope12
slope12*age80*DV   nm5rt     0.01406     0.08268     101        0.17       0.8653  DV5 age on slope26
slope26*age80*DV   nm3rt    -0.01350     0.01076     101       -1.25       0.2125  DV3 age on slope26
slope26*age80*DV   nm5rt    -0.01448     0.01877     101       -0.77       0.4423  DV5 age on slope26

            Type 3 Tests of Fixed Effects
                     Num     Den
Effect                DF      DF    F Value    Pr > F
DV                     2     101     741.34    <.0001  Are both intercepts NE 0?
slope12*DV             2     101      21.43    <.0001  Are both fixed slope12 NE 0?
slope26*DV             2     101      13.87    <.0001  Are both fixed slope26 NE 0?
age80*DV               2     101       6.30    0.0026  Are both age effects on intercept NE 0
slope12*age80*DV       2     101       1.35    0.2636  Are both age effects on slope12 NE 0?
slope26*age80*DV       2     101       0.84    0.4342  Are both age effects on slope26 NE 0?
```

As we requested, these ESTIMATE statements test the difference across DVs in each fixed effect of age.

```
                                   Estimates
                                         Standard
Label                         Estimate      Error      DF    t Value    Pr > |t|
DV Diff in Age Main Effect      0.1386     0.1377     101       1.01      0.3164
DV Diff in Age*Slope12         0.08987    0.08510     101       1.06      0.2934
DV Diff in Age*Slope26        -0.00098    0.01784     101      -0.05      0.9563
```

As we requested, these CONTRAST statements test the 3 age effects and 3 DV differences therein at once.

```
                                        Contrasts
                              Num      Den
Label                          DF       DF    Chi-Square    F Value    Pr > ChiSq    Pr > F
DV3 DF=3 age effect             3      101         12.48       4.16        0.0059    0.0080
DV5 DF=3 age effect             3      101          8.35       2.78        0.0393    0.0448
DV Diff in 3 Age Effects        3      101          7.66       2.55        0.0535    0.0596
```

```
* Pseudo-R2 for age relative to unconditional random slopes model;
%PseudoR2(Ncov=24, CovFewer=CovRand, CovMore=CovAge);
```

```
PsuedoR2 (% Reduction) for CovRand vs. CovAge
 Name      CovParm     Subject              Estimate      StdErr    ZValue     ProbZ     PseudoR2
CovRand    UN(1,1)     PersonID              28.1322      4.2099      6.68     <.0001        .
CovRand    UN(2,2)     PersonID               130.24     19.8749      6.55     <.0001        .
CovRand    UN(3,3)     PersonID               6.3041      1.3054      4.83     <.0001        .
CovRand    UN(4,4)     PersonID               7.8667      3.8230      2.06     0.0198        .
CovRand    UN(5,5)     PersonID               0.2573     0.06273      4.10     <.0001        .
CovRand    UN(6,6)     PersonID               0.2224      0.2042      1.09     0.1380        .
CovRand    UN(1,1)     PersonID*session       1.7673      0.1436     12.31     <.0001        .
CovRand    UN(2,2)     PersonID*session      10.8625      0.8825     12.31     <.0001        .
CovAge     UN(1,1)     PersonID              24.8900      3.7540      6.63     <.0001      0.11525   for L2 DV3 int
CovAge     UN(2,2)     PersonID               123.27     18.8960      6.52     <.0001      0.05347   for L2 DV5 int
CovAge     UN(3,3)     PersonID               6.0940      1.2763      4.77     <.0001      0.03333   for L2 DV3 slope12
CovAge     UN(4,4)     PersonID               7.8595      3.8221      2.06     0.0199      0.00092   for L2 DV3 slope12
CovAge     UN(5,5)     PersonID               0.2507     0.06184      4.05     <.0001      0.02557   for L2 DV3 slope26
CovAge     UN(6,6)     PersonID               0.2147      0.2032      1.06     0.1453      0.03444   for L2 DV5 slope26
CovAge     UN(1,1)     PersonID*session       1.7673      0.1436     12.31     <.0001      0.00000   for L1 DV3 res
CovAge     UN(2,2)     PersonID*session      10.8625      0.8825     12.31     <.0001     -0.00000   for L1 DV5 res
```

```
* Total-R2 relative to empty model -- note I made a new macro for this;
%TotalR2multiv(DV=y, PredFewer=PredTime, PredMore=PredAge);
```

```
Total R2 (% Reduction) for PredTime vs. PredAge
                      Pred                   Total
  Name       DV       Corr      TotalR2     R2Diff
PredTime    nm3rt    0.19338    0.03740        .
PredAge     nm3rt    0.32795    0.10755    0.070156
PredTime    nm5rt    0.08929    0.00797        .
PredAge     nm5rt    0.25967    0.06743    0.059456
```

Sample Results Section:

The extent to which individual differences in response time (RT) over six sessions for simple and complex versions of a processing speed test (number match 3 and 5, or NM3 and NM5, respectively) could be predicted from baseline age was examined in a series of multivariate multilevel models (i.e., general linear mixed models) in which the six practice sessions at level 1 were nested within each participant at level 2, and the two tests were modeled simultaneously as multivariate outcomes. For numeric stability, the two outcomes of response time in milliseconds were divided by 10. All model parameters were estimated separately by outcome, and all possible covariances at each level between outcomes were estimated as well. Maximum likelihood (ML) in SAS PROC MIXED was used to estimate all model parameters; denominator degrees of freedom were estimated using the Satterthwaite method. The significance of new fixed effects were evaluated with univariate and multivariate Wald tests. Effect sizes are reported below using pseudo-R^2, or the proportion reduction in each variance component, as well as total-R^2, the squared correlation between the original outcome and the outcome predicted by the model fixed effects.

Empty means models (i.e., including no predictors) with only random intercepts for each outcome indicated that 81.6% and 75.5% of the variance in NM3 and NM5, respectively, was due to mean differences between persons. At level 2, the random intercepts had a significant covariance across outcomes (r = .898), indicating the individuals who had faster response times on average for NM3 were highly likely to have faster average response times for NM5 as well. Likewise, at level 1, the residuals had a significant covariance across outcomes (r = .407), indicating that on occasions when individuals were faster than their own on average on NM3, they were likely to be faster than their own average on NM5 at that same occasion as well.

Change over time was then modeled using two piecewise linear slopes: *slope12* indicated the rate of change per session between sessions 1 to 2, whereas *slope26* indicated the rate of change per session from sessions 2 to 6. Adding fixed effects for slope12 and slope26 for each outcome reduced the level-1 residual variance by 24.4% for NM3 and by 7.17% for NM5, resulting in total-R^2 = .037 for NM3 and total-R^2 = .008 for NM5. Adding random variances for each of the four slopes (as well as possible random effect covariances) resulted in significant model improvement, $-2\Delta LL(\sim 18) = 149.60$, $p < .001$, indicating that individuals varied significantly in their rates of change. At level 2, the random intercepts had a significant covariance across outcomes (r = .807), indicating the individuals who had faster predicted response times than others at session 1 for NM3 were highly likely to have faster predicted response times than others at session 1 for NM5 as well. Also at level 2, the random slopes for the rate of change per session between sessions 1 and 2 were not significantly related across outcomes (r = .231), indicating that individuals with greater initial improvement than others on NM3 were not necessarily likely to have greater initial improvement than others on NM5 as well. The covariance between the random slopes for the rate of change per session between sessions 2 and 6 were marginally significantly related across outcomes (r = .652), indicating that individuals with greater later improvement than others on NM3 were somewhat likely to have greater later improvement than others on NM5 as well. Finally, at level 1, the residuals retained their significant covariance across outcomes (r = .292), indicating that on occasions when individuals were faster than predicted by their own trajectory on NM3, they were likely to be faster than predicted by their own trajectory on NM5 at that same occasion as well. In examining the fixed effects, there was a significant difference in the fixed intercept for the expected response time at session 1 across outcomes, such that NM5 was slower than NM3 on average. There was a significant overall difference in the slopes for change across sessions across outcomes, $F(2, 101) = 3.16$, $p = .047$. More specifically, slope12 was significantly negative for each outcome, indicating that response time decreased significantly on average between sessions 1 and 2, and there was no difference across outcomes in the fixed effect of slope12. In contrast, slope26 was significantly negative only for NM3, indicating that NM3 was predicted to continue to improve on average after session 2, but NM5 was not. As a result, the fixed effect of slope26 was significantly more negative for NM3.

We then examined age at baseline (centered such that 0 = 80 years) as a predictor of each intercept and piecewise linear slope. The three effects of age together resulted in a significant omnibus effect for both NM3, $F(3, 101) = 4.16$, $p = .008$, and NM5 $F(3, 101) = 2.78$, $p = .045$, and there was a marginally significant difference across outcomes in these overall effects of age, $F(3, 101) = 2.55$, $p = .060$. These effects of age accounted for an additional 7.02% and 5.95% of the total variance in NM3 and NM5, respectively. However, only the fixed effects of age on the intercept was significant, indicating that response time at the first session was predicted to be significantly slower in older persons, equivalently so across outcomes. These simple main effects of age accounted for 11.5% and 5.3% of the level-2 random intercept variance in NM3 and NM5, respectively. The nonsignificant effects of age on slope12 (which were also equivalent across outcomes) accounted for 3.33% and < 1% of the level-2 random slope12 variance in NM3 and NM5, respectively. The nonsignificant effects of age on slope26 (which were also equivalent across outcomes) accounted for 2.56% and 3.44% of the level-2 random slope26 variance in NM3 and NM5, respectively.